Control structures, fifth lecture

## The practice of effects: from exceptions to effect handlers

Xavier Leroy
2024-02-23

Collège de France, chair of software sciences
xavier.leroy@college-de-france.fr

## Exceptions

## Exceptions in a functional language

An exception = a value (type exn) that describes an exceptional condition (error, lack of a meaningful result, ...).

Expressions:

$$
\begin{aligned}
e:: & = & \operatorname{cst}|x| \lambda x . e \mid e_{1} e_{2} & \\
& \mid \text { raise } e & & \text { raising an exception } \\
& \mid \text { try } e_{1} \text { with } x \rightarrow e_{2} & & \text { handling an exception }
\end{aligned}
$$

raise $e$ stops evaluation and branches to the nearest enclosing try... with. This expression returns no value.
(As shown by the type raise : $\forall \alpha$, exn $\rightarrow \alpha$.)

## Exceptions in a functional language

An exception = a value (type exn) that describes an exceptional condition (error, lack of a meaningful result, ...).

Expressions:

$$
\begin{aligned}
e:: & = & \operatorname{cst}|x| \lambda x . e \mid e_{1} e_{2} & \\
& \mid \text { raise } e & & \text { raising an exception } \\
& \mid \text { try } e_{1} \text { with } x \rightarrow e_{2} & & \text { handling an exception }
\end{aligned}
$$

try $e_{1}$ with $x \rightarrow e_{2}$ evaluates the body $e_{1}$.
If $e_{1}$ raises no exception, its value is returned as the value of the whole try...with.
If $e_{1}$ raises an exception, the value $v$ of the exception is bound to $x$ and the handler $e_{2}$ is evaluated.

## Examples of uses of exceptions

Error reporting (for instance, arithmetic overflow):

```
let safe_add x y =
    let z = x + y in
    if (z lxor x) land (z lxor y) < O then raise Overflow;
    z
let sum_list l =
    try
        let s = List.fold_left safe_add 0 l in
        printf "Sum is %d\n" s
        with Overflow ->
        printf "Overflow!\n"
```


## Examples of uses of exceptions

Early exit from nested recursive calls:

```
let list_product l =
    let exception Zero in
    let rec product = function
        | [] -> 1
        | 0 :: _ -> raise Zero
        | n :: l -> n * product l
    in
    try product l with Zero -> 0
```


## Examples of uses of exceptions

Emulating break and continue:

```
exception Break in
exception Continue in
try
        for i = lo to hi do
            try
                                    ... raise Break ... raise Continue ...
            with Continue -> ()
        done
with Break -> ()
```

Exceptions that are raised and handled in the same function $\approx$ multi-level exit (lecture \#1) $\approx$ forward goto.

## Reduction semantics

Two head-reduction rules for try...with:

$$
\begin{array}{r}
\text { try } v \text { with } x \rightarrow e \\
\text { try } D[\text { raise } v] \text { with } x \rightarrow e
\end{array} \begin{aligned}
& \varepsilon \\
& \hline
\end{aligned} e\{x \leftarrow v\}
$$

Here, $D$ is a context with no try... with enclosing the hole:
Reduction contexts:

$$
C::=[]|C e| v C \mid \text { raise } C \mid \operatorname{try} C \text { with } x \rightarrow e
$$

Exception propagation contexts:

$$
D::=[]|D e| v D \mid \text { raise } D
$$

(See later: the semantics of effect handlers here .)

## Reduction semantics

Consider a program $p$ that is about to raise exception $v$ :

$$
p=C[\text { raise } v]
$$

If the raise $v$ is enclosed in a try... with, we write $p$ as

$$
p=C^{\prime}[\text { try } D[\text { raise } v] \text { with } x \rightarrow e]
$$

and we reduce

$$
p \rightarrow C^{\prime}[e\{x \leftarrow v\}]
$$

If the raise $v$ is not enclosed in any try...with, program $p$ is stuck on an uncaught exception.

## Exception-returning style (ERS)

An alternative to exceptions: include errors in the return values of functions.

```
type ('a, 'e) result = V of 'a | E of 'e
let safe_add x y : (int, string) result =
    let z = x + y in
    if (z lxor x) land (z lxor y) < 0
    then E "overflow"
    else V z
let rec safe_add_list = function
    | [] -> V 0
    | x :: l ->
        match safe_add_list l with
        | V y -> safe_add x y
        | E e -> E e
```


## The ERS transformation

$$
\begin{aligned}
& \mathcal{E}(c s t)= V c s t \\
& \mathcal{E}(x)= V x \\
& \mathcal{E}(\lambda x . e)= V(\lambda x \cdot \mathcal{E}(e)) \\
& \mathcal{E}\left(e_{1} e_{2}\right)= \text { match } \mathcal{E}\left(e_{1}\right) \text { with } E x_{1} \rightarrow E x_{1} \mid V v_{1} \rightarrow \\
& \text { match } \mathcal{E}\left(e_{2}\right) \text { with } E x_{2} \rightarrow E x_{2} \mid V v_{2} \rightarrow v_{1} v_{2} \\
& \mathcal{E}(\text { raise } e)= \text { match } \mathcal{E}(e) \text { with } E x \rightarrow E x \mid V v \rightarrow E v \\
& \mathcal{E}\left(\operatorname{try} e_{1} \text { with } x \rightarrow e_{2}\right) \\
&= \text { match } \mathcal{E}\left(e_{1}\right) \text { with } E x \rightarrow \mathcal{E}\left(e_{2}\right) \mid V v \rightarrow V v
\end{aligned}
$$

The transformation propagates error results "upward", except for try....with, which handles the error result.

## Alternative: "double-barreled" CPS

Two continuations: k 1 to return a value, k 2 to raise an exception.

```
let safe_add x y k1 k2 =
    let z = x + y in
    if (z lxor x) land (z lxor y) < 0
    then k2 "overflow"
    else k1 z
let rec safe_add_list l k1 k2 =
    match l with
    | [] -> k1 0
    | x :: l ->
        safe_add_list l (fun v -> safe_add x v k1 k2) k2
```


## A double-barreled CPS transformation

$$
\begin{aligned}
\mathcal{C}^{2}(c s t) & =\lambda k_{1} \cdot \lambda k_{2} \cdot k_{1} \text { cst } \\
\mathcal{C}^{2}(x) & =\lambda k_{1} \cdot \lambda k_{2} \cdot k_{1} x \\
\mathcal{C}^{2}(\lambda x \cdot e) & =\lambda k_{1} \cdot \lambda k_{2} \cdot k_{1}\left(\lambda x \cdot \mathcal{C}^{2}(e)\right) \\
\mathcal{C}^{2}\left(e_{1} e_{2}\right) & =\lambda k_{1} \cdot \lambda k_{2} \cdot \mathcal{C}^{2}\left(e_{1}\right)\left(\lambda v_{1} \cdot \mathcal{C}^{2}\left(e_{2}\right)\left(\lambda v_{2} \cdot v_{1} v_{2} k_{1} k_{2}\right) k_{2}\right) k_{2} \\
\mathcal{C}^{2}(\text { raise } e) & =\lambda k_{1} \cdot \lambda k_{2} \cdot \mathcal{C}^{2}(e) k_{2} k_{2} \\
\mathcal{C}^{2}\left(\operatorname{try} e_{1}\right. \text { with } & \left.x \rightarrow e_{2}\right) \\
& =\lambda k_{1} \cdot \lambda k_{2} \cdot \mathcal{C}^{2}\left(e_{1}\right) k_{1}\left(\lambda x \cdot \mathcal{C}^{2}\left(e_{2}\right) k_{1} k_{2}\right)
\end{aligned}
$$

The transformation propagates the error continuation $k_{2}$ "downward" (towards sub-expressions), except for try...with, which installs a new error continuation.

## Double-barreled CPS transformation

## $\approx$ ERS transformation followed by CPS transformation

For a program of a base type $\tau$ :


Same type isomorphism as $(A+B) \rightarrow C \approx(A \rightarrow C) \times(B \rightarrow C)$.

## Effects and effect handlers

## Effects and effect handlers

Algebraic effects:
(Plotkin, Power, Pretnar, 2003, 2009)
A theory of the generation, propagation and specification of effects in programming languages.
(Effects = mutable state, I/O, exceptions, non-determinism, ...). ( $\rightarrow$ Lecture \#6)

User-defined effects and effect handlers: (Bauer \& Pretnar, 2015)
A powerful control structure inspired by the theory of algebraic effects.

Combines restartable exceptions with delimited continuations.

## Catching errors using exceptions

```
type exn += Conversion_failure of string
let int_of_string s =
    match int_of_string_opt s with
    | Some n -> n
    | None -> raise (Conversion_failure s)
let sum_stringlist lst =
    lst |> List.map int_of_string |> List.fold_left (+) 0
let safe_sum_stringlist lst =
    match sum_stringlist lst with
    | res -> res
    | exception Conversion_failure s ->
        printf "Bad input: %s\n" s; max_int
```


## Fixing errors using effects

```
type _ eff += Conversion_failure : string -> int eff
let int_of_string s =
    match int_of_string_opt s with
    | Some n -> n
    | None -> perform (Conversion_failure s)
let sum_stringlist lst =
    lst |> List.map int_of_string |> List.fold_left (+) 0
let safe_sum_stringlist lst =
    match sum_stringlist lst with
    | res -> res
    | effect Conversion_failure s, k ->
        printf "Bad input: %s, replaced with 0\n" s;
        continue k 0
```


## Example of execution

Without the effect handler: behaves like an uncaught exception.
\# let n = sum_stringlist ["1"; "xxx"; "2"; "yyy"]
Exception: Stdlib.Effect.Unhandled(Conversion_failure("xxx"))
With the effect handler: errors are caught and fixed.

```
# let n = safe_sum_stringlist ["1"; "xxx"; "2"; "yyy"]
Bad input xxx, replaced with 0
Bad input yyy, replaced with 0
val n : int = 3
```

(Examples written and run in OCaml 5.1.1 + an experimental syntax match with effect. To use: opam switch create 5.1.1+effect-syntax .)

## Effects and continuations

```
let int_of_string s = ... perform (Conversion_failure s)
let safe_sum_stringlist lst =
    match ...
    with effect Conversion_failure s, k -> ... continue k 0
```

When perform raises an effect, its (delimited) continuation is captured and given to the handler along with the effect value.

The effect handler can either discard this continuation k , or restart it on a value of the type expected by the context of the perform (here, int).

Limitation (in OCaml, not in other languages): the continuation is "one-shot" (linear) and must be restarted or discarded exactly once.

## Intuitions in terms of call stacks

Raising an exception = cutting the stack.


## Intuitions in terms of call stacks

Raising an exception = cutting the stack.


## Intuitions in terms of call stacks

Raising an exception = cutting the stack.


## Intuitions in terms of call stacks

Raising an exception = cutting the stack.


## Intuitions in terms of call stacks

Raising an exception = cutting the stack.


## Intuitions in terms of call stacks

Naive undelimited continuations = stack copies (to the heap).


## Intuitions in terms of call stacks

Naive undelimited continuations = stack copies (to the heap).


## Intuitions in terms of call stacks

Naive undelimited continuations = stack copies (to the heap).


## Intuitions in terms of call stacks

Naive undelimited continuations = stack copies (to the heap).


## Intuitions in terms of call stacks

Naive undelimited continuations = stack copies (to the heap).


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


## Intuitions in terms of call stacks

Effect handling = switching between several stacks.


In OCaml: no stack copying $\rightarrow$ one-shot continuations.

## Deep handlers, shallow handlers

## Deep handler:

remains in place when a continuation is restarted;
disappears only when the computation terminates normally.

```
# let n = safe_sum_stringlist ["1"; "xxx"; "2"; "yyy"]
Bad input xxx, replaced with 0
Bad input yyy, replaced with 0
val n : int = 3
```


## Shallow handler:

disappears as soon as an effect is handled.

```
# let n = safe_sum_stringlist ["1"; "xxx"; "2"; "yyy"]
Bad input xxx, replaced with 0
Exception: Stdlib.Effect.Unhandled(Conversion_failure("yyy"))
```

(In OCaml: match...with is "deep"; the Effect.Shallow library implements the "shallow" semantics.)

## Control inversion on an iterator

As in lecture \#4, we assume given an "internal" iterator such as the one over binary trees:

```
type 'a tree = Leaf | Node of 'a tree * 'a * 'a tree
let rec tree_iter (f: 'a -> unit) (t: 'a tree) =
    match t with
    | Leaf -> ()
    | Node(l, x, r) -> tree_iter f l; f x; tree_iter f r
```

We'd like to implement an "external" iterator on top of tree_iter:

```
type 'a enum = Done | More of 'a * (unit -> 'a enum)
val tree_enum : 'a tree -> 'a enum
```


## Control inversion on an iterator

```
let tree_enum (type elt) : elt tree -> elt enum =
    let module Inv = struct
        type _ eff += Next : elt -> unit eff
        let tree_enum (t: elt tree) : elt enum =
        match tree_iter (fun x -> perform (Next x)) t with
        | () -> Done
        | effect Next x, k -> More(x, fun () -> continue k ())
    end in
    Inv.tree_enum
```

We use OCaml's local modules to declare an effect Next that is local to the function and has the right type to make tree_enum polymorphic in the type elt of elements.

## Control inversion on an iterator

```
let tree_enum (type elt) : elt tree -> elt enum =
    let module Inv = struct
        type _ eff += Next : elt -> unit eff
        let tree_enum (t: elt tree) : elt enum =
        match tree_iter (fun x -> perform (Next x)) t with
        | () -> Done
        | effect Next x, k -> More(x, fun () -> continue k ())
    end in
    Inv.tree_enum
```

For each element x of the tree, the effect Next x is performed. The handler receives x and the continuation k that restarts the traversal.

## Control inversion on an iterator

```
let tree_enum (type elt) : elt tree -> elt enum =
    let module Inv = struct
        type _ eff += Next : elt -> unit eff
        let tree_enum (t: elt tree) : elt enum =
        match tree_iter (fun x -> perform (Next x)) t with
        | () -> Done
        | effect Next x, k -> More(x, fun () -> continue k ())
    end in
    Inv.tree_enum
```

When the traversal is over, tree_iter returns (), which is turned into Done by the effect handler.

## Control inversion on an iterator

let tree_enum (type elt) : elt tree -> elt enum =
let module Inv = struct

$$
\begin{aligned}
& \text { type _ eff += Next : elt -> unit eff } \\
& \text { let tree_enum (t: elt tree) : elt enum = } \\
& \text { match tree_iter (fun x -> perform (Next x)) t with } \\
& \text { | () -> Done } \\
& \text { | effect Next } x, k \text { More(x, fun () }->\text { continue k ()) }
\end{aligned}
$$

end in
Inv.tree_enum

Note that the handler changes the type of the computation:
tree_iter ... t has type unit,
match tree_iter ... has type elt enum.

## Comparing callcc with effect handling

```
Using callcc:(lecture #4)
callcc (fun k ->
    tree_iter
    (fun x ->
        callcc
            (fun k' ->
                        k (More(x, k'))))
    t;
    Done)
```

Two callcc: one to exit, one to support restarting. More (x, ...) is computed in the iterated function.

## Using effect handling:

```
match
    tree_iter
        (fun x -> perform (Next x))
        t
with
| () -> Done
| effect Next x, k ->
    More(x, fun () -> resume k ())
```

A single perform to exit while capturing the restart continuation.

More (x, . . .) is computed in the handler.

## Control inversion on an iterator

This construction can be generalized to invert any internal iterator on any collection type:
let enum_of_iter
(type elt) (type collection)
(iter: (elt -> unit) -> collection -> unit)
: collection -> elt enum =
let module Inv = struct
type _ eff += Next : elt -> unit eff
let enum coll =
match iter (fun x $\rightarrow$ perform (Next x)) coll with
| () -> Done
| effect Next x, k -> More(x, fun () -> continue k ())
end in Inv.enum

## Transforming and re-emitting effects

(M. Pretnar, An introduction to algebraic effects and handlers, 2015.)

An effect Print for outputting a string.
type _ eff += Print : string -> unit eff
let print s = perform (Print s)
let abc () = print "a"; print "b"; print "c"

## Transforming and re-emitting effects

The effect can be handled as a "true" output on the terminal:
let output $f=$
match f () with
| () -> print_newline()
| effect Print s, k $\rightarrow$ print_string s; continue k ()
But we can also collect all outputs in a string:
let collect $\mathrm{f}=$
match f () with
| () -> ""
| effect Print s, k -> s ~ continue k ()
collect abc produces the string "abc".

## Transforming and re-emitting effects

We can also re-emit the Print effect after processing it, for instance to reverse the order of outputs:

```
let reverse f =
match f () with
    | () -> ()
    | effect Print s, k -> continue k (); print s
```

or to add a sequence number:
let number $f=$
begin match f () with
| () -> (fun lineno -> ())
| effect Print s, k ->
(fun lineno ->
print (sprintf "\%d:\%s ${ }^{n}$ " lineno s);
continue $k$ () (lineno + 1))
end 1

# Implementing cooperative threads with effects and handlers 

## A library for cooperative threading

The natural interface in "direct style":
spawn: (unit -> unit) -> unit
Start a new thread.
yield: unit -> unit
Suspend the current thread;
switch to another runnable thread.
terminate: unit -> unit
Stop the current thread forever.

## Defining the corresponding effects

The three operations are defined trivially as raising effects (which will be handled by the scheduler).

```
type _ eff +=
    | Spawn : (unit -> unit) -> unit eff
    | Yield : unit eff
    | Terminate : unit eff
let spawn f = perform (Spawn f)
let yield () = perform Yield
let terminate () = perform Terminate
```


## The state of the scheduler

A queue of threads that were suspended by a call to yield, ready to be restarted.

```
let runnable : (unit -> unit) Queue.t = Queue.create()
let suspend f = Queue.add f runnable
let restart () =
    match Queue.take_opt runnable with
    | None -> ()
    | Some f -> f ()
```


## The scheduler

let rec run (f: unit -> unit) =
match $f()$ with
| () -> restart ()
| effect Terminate, k -> discontinue k; restart ()
| effect Yield, k -> suspend (continue k); restart ()
| effect Spawn f, k $\rightarrow$ suspend (continue k) ; run $f$

## The scheduler

```
let rec run (f: unit -> unit) =
    match f() with
    | () -> restart ()
    | effect Terminate, k -> discontinue k; restart ()
    | effect Yield, k -> suspend (continue k); restart ()
    | effect Spawn f, k -> suspend (continue k); run f
```

The current thread terminates normally: we restart another thread.

## The scheduler

```
let rec run (f: unit -> unit) =
    match f() with
    | () -> restart ()
    | effect Terminate, k -> discontinue k; restart ()
    | effect Yield, k -> suspend (continue k); restart ()
    | effect Spawn f, k -> suspend (continue k); run f
```

The current thread called terminate:
we "discontinue" (throw away) the continuation $k$ (the thread will never restart) and we restart another thread.

## The scheduler

```
let rec run (f: unit -> unit) =
    match f() with
    | () -> restart ()
    | effect Terminate, k -> discontinue k; restart ()
    | effect Yield, k -> suspend (continue k); restart ()
    | effect Spawn f, k -> suspend (continue k); run f
```

The current thread called yield: we store the continuation k as ready to restart, and we restart another thread.

## The scheduler

```
let rec run (f: unit -> unit) =
    match f() with
    | () -> restart ()
    | effect Terminate, k -> discontinue k; restart ()
    | effect Yield, k -> suspend (continue k); restart ()
    | effect Spawn f, k -> suspend (continue k); run f
```

The current thread called spawn f : we store the continuation k as ready to restart, and we start to execute f.

## The scheduler

let rec run (f: unit -> unit) = match $f()$ with
| () -> restart ()
| effect Terminate, k -> discontinue k; restart ()
| effect Yield, k -> suspend (continue k); restart ()
| effect Spawn f, k -> suspend (continue k) ; run f

Alternative:

```
| effect Spawn f, k ->
    suspend (fun () -> run f); continue k ()
```

In both cases, we must do run f, and not just f(), so that the effects of $f()$ are handled.

## Example of use

A client of the library, written in direct style:
let task name $\mathrm{n}=$ for $i=1$ to $n$ do printf " $\%$ s\%d " name i; yield() done
let _ =
run (fun () ->
spawn (fun () -> task "a" 6); spawn (fun () -> task "b" 3); task "c" 4)

Prints a1 b1 a2 c1 b2 a3 c2 b3 a4 c3 a5 c4 a6

## Adding message-passing communication

new_channel: unit -> 'a channel
Create a new channel to pass values of type 'a.
recv: 'a channel -> 'a
Receive a message from the given channel.
send: 'a channel -> 'a -> unit
Send the given message on the given channel.
We choose to implement "rendez-vous" semantics ( $\pi$-calculus):
send ch v blocks until another thread calls recv ch;
both threads restart;
recv ch returns value v.

## Structure of a communication channel

A channel = two queues, one for threads blocked on a send waiting for a matching recv, the other for threads blocked on a recv waiting for a send.

```
type 'a channel = {
    senders: ('a * (unit, unit) continuation) Queue.t;
    receivers: ('a, unit) continuation Queue.t
    }
let new_channel () =
    { senders = Queue.create(); receivers = Queue.create() }
```

At any time, at least one of the two queues is empty.

## Message-sending operations

As always, whenever we have operations that cannot be implemented locally and must be handled by the scheduler, we turn these operators into effects.
type _ eff +=
| Send : 'a channel * 'a -> unit eff
| Recv : 'a channel -> 'a eff
let send ch v = perform (Send (ch, v))
let recv ch = perform (Recv ch)

## The scheduler extended with message passing

```
let rec run (f: unit -> unit) =
    match f () with
    . . .
    | effect Send(ch, v), k ->
        begin match Queue.take_opt ch.receivers with
        | Some rc -> suspend (continue k) ; continue rc v
        | None -> Queue.add (v, k) ch.senders; restart()
        end
    | effect Recv ch, k ->
        begin match Queue.take_opt ch.senders with
        | Some(v, sn) -> suspend (continue sn); continue k v
        | None -> Queue.add k ch.receivers; restart()
        end
```


## Semantics of effect handlers

## A small functional languages with effects and handlers

Expressions:

$$
\begin{aligned}
e:: & = & \operatorname{cst}|x| \lambda x \cdot e \mid e_{1} e_{2} & \\
& \mid \text { perform } e & & \text { perform effect } e \\
& \mid \text { handle } e \text { with } e_{\text {ret }}, e_{\text {eff }} & & \text { handle effects in } e
\end{aligned}
$$

perform e stops evaluation and branches to the nearest enclosing handle.

## A small functional languages with effects and handlers

Expressions:

$$
\begin{aligned}
e:: & = & \operatorname{cst}|x| \lambda x . e \mid e_{1} e_{2} & \\
& \mid \text { perform } e & & \text { perform effect } e \\
& \mid \text { handle } e \text { with } e_{\text {ret }}, e_{\text {eff }} & & \text { handle effects in } e
\end{aligned}
$$

handle $e$ with $e_{\text {ret }}, e_{\text {eff }}$ evaluates the body $e$.
If $e$ evaluates to value $v$ without performing effects, we apply $e_{\text {ret }}$ to $v$.

If $e$ performs effect $f$, we apply $e_{\text {eff }}$ to $(f, k)$
where $f$ is the value of the effect and $k$ the continuation of the perform.

## Encoding match...with effect

Adding extensible algebraic datatypes and pattern-matching, we can encode

$$
\begin{aligned}
& \text { match } e \text { with } \\
& \mid x \rightarrow e_{0} \\
& \mid \text { effect } F_{1} x_{1}, k \rightarrow e_{1} \\
& \vdots \\
& \mid \text { effect } F_{n} x_{n}, k \rightarrow e_{n}
\end{aligned}
$$

as
handle $e$ with
$\left(\lambda x . e_{0}\right)$,
$(\lambda(f, k)$. match $f$ with
$\left|F_{1} x_{1} \rightarrow e_{1}\right| \ldots \mid F_{n} x_{n} \rightarrow e_{n}$
$\left.\right|_{-} \rightarrow k($ perform $\left.f)\right)$

## Reduction semantics

(Very close to the reduction semantics for exceptions here.)
Two head-reduction rules forhandle:

$$
\begin{array}{r}
\text { handle } v \text { with } e_{1}, e_{2} \xrightarrow{\varepsilon} e_{1} v \\
\text { handle } D[\text { perform } v] \text { with } e_{1}, e_{2} \xrightarrow{\varepsilon} e_{2}\left(v,\left(\lambda v^{\prime} . D\left[v^{\prime}\right]\right)\right)
\end{array}
$$

Here, $D$ is a context with no handle enclosing the hole:
Reduction contexts:

$$
C::=[]|C e| v C \mid \text { perform } C \mid \text { handle } C \text { with } e_{1}, e_{2}
$$

Effect propagation contexts:

$$
D::=[]|D e| v D \mid \text { perform } D
$$

## Deep handlers, shallow handlers

$$
\begin{aligned}
& \text { handle } D[\text { perform } v] \text { with } e_{1}, e_{2} \\
&\left.\xrightarrow{\&} e_{2}\left(v, \lambda v^{\prime} . D\left[v^{\prime}\right]\right)\right)
\end{aligned}
$$

The rule above implements shallow handling: the handler is no longer active when the continuation $D$ is restarted.

Deep handling is obtained by reinstalling the handler around the continuation $D$ :

$$
\begin{aligned}
& \text { handle } D[\text { perform } v] \text { with } e_{1}, e_{2} \\
& \quad \stackrel{\varepsilon}{\rightarrow} e_{2}\left(v, \lambda v^{\prime} \text {. handle } D\left[v^{\prime}\right] \text { with } e_{1}, e_{2}\right)
\end{aligned}
$$

## CPS transformation for delimited continuations

(M. Materzok, D. Biernacki, Subtyping delimited continuations, 2011.)

For undelimited continuations (callcc), a CPS-transformed term takes a continuation $k$ as argument, and ensures that

$$
\mathcal{C}(e) k \xrightarrow{*} k c s t \quad \text { if } e \xrightarrow{*} c s t
$$

For delimited continuations, a CPS-transformed term takes $n+1$ continuations $k_{0}, \ldots, k_{n}$ as arguments, where $n$ is the number of enclosing delimiters, and each $k_{i}$ is the continuation up to the next delimiter.

$$
\mathcal{C}(e) k_{0} k_{1} \ldots k_{n} \xrightarrow{*} k_{0} \operatorname{cst} k_{1} \ldots k_{n} \quad \text { if } \quad e \xrightarrow{*} \operatorname{cst}
$$

## CPS transformation for the pure subset of the language

$$
\begin{aligned}
\mathcal{C}(c s t) & =\lambda k \cdot k c s t \\
\mathcal{C}(x) & =\lambda k \cdot k x \\
\mathcal{C}(\lambda x \cdot e) & =\lambda k \cdot k(\lambda x \cdot \mathcal{C}(e)) \\
\mathcal{C}\left(e_{1} e_{2}\right) & =\lambda k \cdot \mathcal{C}\left(e_{1}\right)\left(\lambda v_{1} \cdot \mathcal{C}\left(e_{2}\right)\left(\lambda v_{2} \cdot v_{1} v_{2} k\right)\right)
\end{aligned}
$$

Same definitions as for the usual CBV-value CPS transformation. These definitions remain correct when $\mathcal{C}(e)$ is applied to $n$ continuations, e.g.

$$
\mathcal{C}(c s t) k_{0} k_{1} \ldots k_{n}=(\lambda k . k c s t) k_{0} k_{1} \ldots k_{n} \rightarrow k_{0} \operatorname{cst} k_{1} \ldots k_{n}
$$

## CPS transformation for delimited continuations

We formalize the operators shift ${ }_{0}$ and reset $_{0}$
(0. Danvy and A. Filinksi, 1989).

A delimiter adds a trivial continuation at the head of the list:

$$
\mathcal{C}(\operatorname{delim} e)=\mathcal{C}(e)(\lambda x \cdot \lambda k \cdot k x)
$$

so that, in the case where $e \xrightarrow{*}$ cst,

$$
\begin{aligned}
\mathcal{C}(\operatorname{delim} e) k_{0} k_{1} \ldots k_{n} & =\mathcal{C}(e)(\lambda x \cdot \lambda k \cdot k x) k_{0} \ldots k_{n} \\
& \xrightarrow{*}(\lambda x \cdot \lambda k \cdot k x) \operatorname{cst} k_{0} \ldots k_{n} \\
& \rightarrow k_{0} \operatorname{cst} k_{1} \ldots k_{n}
\end{aligned}
$$

## CPS transformation for delimited continuations

Symmetrically, the capture operator reifies the first continuation to a value, and removes it from the list:

$$
\mathcal{C}(\text { capture }(\lambda k . e))=\lambda k . \mathcal{C}(e)
$$

so that

$$
\mathcal{C}(\operatorname{capture}(\lambda k . e)) k_{0} k_{1} \ldots k_{n}=\mathcal{C}(e)\left[k \leftarrow k_{0}\right] k_{1} \ldots k_{n}
$$

The evaluation of $e$ continues with $k_{1}$, the continuation "after" the nearest delimiter.

The continuation up to this delimiter, $k_{0}$, is captured as the $k$ parameter to $e$.

## CPS transformation for effect handlers

(D. Hillerström, S. Lindley, R. Atkey, Effect handlers via generalised continuations, 2020.)

The previous approach + the "double-barreled" approach: a CPS-transformed term takes $2 n+2$ continuations as arguments, with $n=$ number of enclosing effect handlers.

$$
\mathcal{C}(e) k_{0} h_{0} k_{1} h_{1} \ldots k_{n} h_{n}
$$

The $k_{0}, \ldots k_{n}$ delimited continuations are invoked to return values as results.

The $h_{0}, \ldots h_{n}$ delimited continuations are invoked to perform effects.

## CPS transformation for effects

For the pure subset of the language: we apply the usual CBV CPS transformation rules.

To perform an effect:

$$
\mathcal{C}(\text { perform } e)=\mathcal{C}(e)(\lambda f . \lambda k . \lambda h . h(f, \lambda x . k x h))
$$

$e$ is evaluated to an effect value $f$.
We capture the normal continuation $k$, as well as the effect continuation $h$, and we invoke $h$, giving it $f$ as the effect value and $k^{\prime}=\lambda x . k \times h$ as the way to resume after perform.
(The application of $k$ to $h$ implements deep handling!)

## CPS transformation for effects

An effect handler adds a normal continuation and an effect continuation:

$$
\mathcal{C}\left(\text { handle } e \text { with } e_{1}, e_{2}\right)=\mathcal{C}(e)\left(\lambda v . \lambda h . \mathcal{C}\left(e_{1}\right) v\right) \mathcal{C}\left(e_{2}\right)
$$

In the case where $e \xrightarrow{*} c s t$,

$$
\begin{aligned}
& \mathcal{C}\left(\text { handle } e \text { with } e_{1}, e_{2}\right) k_{0} h_{0} \ldots k_{n} h_{n} \\
& \quad=\mathcal{C}(e)\left(\lambda v \cdot \lambda h \cdot \mathcal{C}\left(e_{1}\right) v\right) \mathcal{C}\left(e_{2}\right) k_{0} h_{0} \ldots k_{n} h_{n} \\
& \quad \xrightarrow{*}\left(\lambda v \cdot \lambda h \cdot \mathcal{C}\left(e_{1}\right) v\right) \operatorname{cst} \mathcal{C}\left(e_{2}\right) k_{0} h_{0} \ldots k_{n} h_{n} \\
& \quad \xrightarrow{*} \mathcal{C}\left(e_{1}\right) \operatorname{cst} k_{0} h_{0} \ldots k_{n} h_{n}
\end{aligned}
$$

In the case where e performs effect $f$ with continuation $k_{f}$, the continuation $C\left(e_{2}\right)$ is applied to $\left(f, k_{f}\right)$ and to the list $k_{0} h_{0} \ldots$

## Summary

## Summary

Effect handlers provide:

- A control operator that supports programming in direct style with delimited continuations.
- A presentation of delimited control as restartable exceptions, more intuitive than the control operators viewed earlier.
- A new programming style: user code performs effects to invoke the services they need; these services are realized by an enclosing handler.


## References

## References

The OCaml version used for the programming examples: opam update \&\& opam switch create 5.1.1+effect-syntax A general introduction to effect handlers:

- Matija Pretnar: An Introduction to Algebraic Effects and Handlers, ENTCS 319, 2015. https://doi.org/10.1016/j.entcs.2015.12.003

CPS transformations for effects:

- Daniel Hillerström, Sam Lindley, Robert Atkey: Effect Handlers via Generalised Continuations, J. Funct. Program. 30, 2020.
https://doi.org/10.1017/S0956796820000040
The implementation of effects in OCaml version 5:
- KC Sivaramakrishnan, Stephen Dolan, Leo White, Tom Kelly, Sadiq Jaffer, Anil Madhavapeddy: Retrofitting Effect Handlers onto OCaml, PLDI, 2021. https://arxiv.org/abs/2104.00250

